DECUS
PROGRAM LIBRARY

DECUS NO.	12-69
TITLE	AN ON-LINE FOCAL-12 PROGRAM FOR AUTO-ANALYZERS
AUTHOR	Mack W. Overton, Jr., Larry L. Alber and Dr. Donald E. Smith
COMPANY	U. S. Food and Drug Administration Chicago, Illinois
DATE	March 10, 1972
SOURCE LANGUAGE	FOCAL-12

AN ON-LINE FOCAL-12 PROGRAM FOR AUTO-ANALYZERS

DECUS Program Library Write-up DECUS NO. 12-69

A program and hardware interface have been developed to acquire and reduce data from an auto-analyzer. A system diagram is presented in Fig. 1. A PDP-12A (8K core) was used for the project. The interfacing circuitry is shown in Fig. 2 and is based on the use of a retransmitting slidewire mounted on the pen cable pulley of the spectrophotometer recorder. The signal taken off the center tap of the slidewire is then conditioned as shown and fed to the A/D converter inputs of the PDP-12A.

The program is written in a modified version of FOCAL-12 (1) and utilizes FOCAL-12 library capabilities for program and user storage. The program is contained in four segments on magnetic tape unit Ø. The first segment is loaded via teletype command and the remaining segments are loaded under program control. Two files on magnetic tape unit 1 provide storage for user input and a maximum of 70 readings (samples and standards) taken from the auto-analyzer. Although storage for a maximum of 70 readings is provided, the first two and the final reading are ignored by the calculation routines. Therefore, in order to utilize fully the storage provided, some program modification would have to be made.

%ONITA1, the first segment, is an initialization routine which is loaded via the library go command issued from the teletype by the user. It is during initialization that basic information concerning the run is supplied by the user. This is accomplished through a user interactive scope-teletype routine.

%ONITA2, the second segment, is loaded under program control upon completion of %ONITA1 and consists of the realtime sample, display and peak hunt routines. A sense switch (sns 1) on the computer console controls the exit and entry into the peak hunt routine. Only the X and Y coordinates for peaks, initial baseline and final baseline are stored. Raw data are not stored.

When sampling is complete, %ONITA3 automatically calculates and outputs the results. At this point the program determines if it found the specified number of peaks, and the user is asked to verify peak identity. If an error is detected %ONITA4 is loaded and the user must code the absorbance values as the computer outputs them by typing S (standard), U(unknown), C(composite), or X (deletion). It should be noted that while %ONITA3 uses an average standard value, %ONITA4 calculates using individual standards (i.e., the most recent standard). Baseline correction is accomplished by fitting a point to a straight line constructed between the initial and final baseline points. This value is subtracted from the corresponding absorbance prior to calculation.

---

[1] FOCAL-12 with overlay to permit reading of sense switches on the PDP-12A console. Courtesy of Mr. Dale L. Lewellyn, Software Consultant, Digital Equipment Corporation, 1350 Frontage Road, Northbrook, Illinois.

A versatile inexpensive system utilizing the PDP-12A has been developed for on-line data acquisition and reduction for auto-analyzers. While admittedly deficient in two areas, no provision for raw data storage or time sharing, it should find use in projects of limited duration which prohibit development of a more sophisticated but expensive system. It also serves admirably as an interim measure during the development of a more sophisticated system.

NOTE

The use of equipment described in this paper does not constitute endorsement by the Food and Drug Administration.



FIG. 1. SYSTEM SCHEMATIC



FIG. 1. Interface cicuit constructed with Burr-Brown Model 603 operational amplifier manifold and plug in components. Burr-Brown Research Corp., Tuscon, Arizona 85706

```
*L L,%ONITA1,Ø
*W A
C FOCAL-12

Ø1.Ø2  L   M,1,ITA,1;L O,F1,F,ITA,1:F I=Ø,84;S F1(I)=Ø
Ø1.Ø4  L   M,1,ITA2,1;L O,F3,F,ITA2,1;F I=Ø,84;S F3(I)=Ø
Ø1.Ø6  O   C; O S
Ø1.Ø7  A   !,"CH NO",F1(Ø)
Ø1.Ø9  A   !,"NO OF ITAS, EXCLUDE COMP",F3(7Ø)
Ø1.1Ø  A   !,"NO PEAKS,SPLS+STDS+COMP",F3(71)
Ø1.11  A   !,"COMP CUP NO.",F3(72)
Ø1.12  O   C;DO 2;A !,"STD CONC",F3(8Ø)
Ø1.13  A   !,"DIL FAC",F3(81)
Ø1.14  A   !,"DECLARED IN MG OR GR?",MG
Ø1.15  I   (MG-ØMG)1.17,1.16,1.14
Ø1.16  S   F3(82)=1;GOTO 1.18
Ø1.17  S   F3(82)=1/64.8
Ø1.18  A   !,"AMT DECLARED",F3(74)
Ø1.19  A   !,"WT OF COMP",F3(83)
Ø1.2Ø  A   !,"AVG TAB WT",F3(84)
Ø1.21  S   F3(73)=F3(8Ø)*F3(81)*F3(82)
Ø1.22  S   F3(75)=F3(83)/F3(84)
Ø1.23  L   C,F1;L C,F2;L C,F3
Ø1.24  L   G,%ONITA2,Ø
Ø1.3Ø  T   "ERROR"
Ø1.31  Q

Ø2.Ø3  A   !,"NO. OF STDS",F1(7Ø)
Ø2.Ø4  T   !,"STD  CUP NO"
Ø2.Ø5  F   I=1,F1(7Ø);DO 3.Ø2
Ø2.Ø6  A   !,"STD DATA OK,Y OR N?",Y
Ø2.Ø7  I   (Y-ØY)2.Ø8,2.Ø9,2.Ø8
Ø2.Ø8  O   C;GOTO 2.Ø3
Ø2.Ø9  O   C

Ø3.Ø2  T   !,I,"      ";A F1(I+7Ø)
*
```

```
*L L,%ONITA2,Ø
*W A
C FOCAL-12

Ø1.Ø1  A    "T",T;O C
Ø1.Ø2  L    O,F1,F,ITA,1;L O,F3,F,ITA2,1
Ø1.Ø3  O    C;O I,T;S J=Ø;S K=Ø;S P=Ø;S M=Ø;S F3(81)=-1
Ø1.Ø4  DO 2;DO 2
Ø1.Ø5  I    (FX(1))1.4Ø,1.Ø4,1.Ø6
Ø1.Ø6  S    F3(77)=<F3(83)+F3(84)>/2;DO 1.Ø3
Ø1.Ø7  I    (FX(1)-1)1.3Ø;DO 2
Ø1.Ø8  I    (F3(84)-F3(83))1.1Ø,1.1Ø
Ø1.Ø9  S    P=P+1;I (P-5)1.11,1.12,1.4Ø
Ø1.1Ø  S    P=Ø
Ø1.11  GOTO 1.Ø7
Ø1.12  DO 1.22;S F3(81)=F3(84)
Ø1.14  I    (FX(1)-1)1.3Ø;DO 2
Ø1.16  I    (F3(84)-F3(83))1.17,1.18,1.18
Ø1.17  S    M=M+1;I (M-5)1.19,1.2Ø,1.4Ø
Ø1.18  S    M=Ø
Ø1.19  GOTO 1.14
Ø1.2Ø  S    F3(76)=F3(76)+1
Ø1.21  S    J=J+1;S F1(J)=F3(82);S F3(J)=F3(81)
Ø1.22  S    P=Ø;S M=Ø
Ø1.23  GOTO 1.Ø7
Ø1.3Ø  S    F3(78)=K;S F3(79)=<F3(83)+F3(84)>/2
Ø1.33  L    C,F1;L C,F3;L G,%ONITA3,Ø
Ø1.4Ø  T    "ERR";Q

Ø2.Ø1  S    K=K+(1Ø*T)
Ø2.Ø2  S    F3(83)=F3(84)
Ø2.Ø3  S    Q=Ø;F I=1,1Ø;S Q=Q+FADC(F1(Ø));O I
Ø2.Ø4  S    F3(84)=Q/1Ø
Ø2.Ø5  I    (F3(84)-F3(81))2.Ø6;S F3(81)=F3(84);S F3(82)=K
Ø2.Ø6  S    O=FDIS(K/(1ØØØØ*T),F3(84))
*
```

```
*L L,%ONITA3,Ø
*W A
C FOCAL-12

Ø1.Ø3  O  T
Ø1.Ø4  L  O,F1,F,ITA,1
Ø1.Ø5  L  O,F3,F,ITA2,1
Ø1.Ø9  S  M=<F3(79)-F3(77)>/F3(78);S ST=Ø
Ø1.1Ø  F  I=73,7Ø+F1(7Ø)-1;S ST=ST+F3(F1(I))-(<M*F1(F1(I))>+F3(77))
Ø1.12  S  ST=ST/<F1(7Ø)-3>
Ø1.14  T  !,"SECONDS  NET  PH  CODE  FOUND  % DECLARED",!,!
Ø1.16  S  I=Ø;S SN=71;S QA=Ø
Ø1.17  S  I=I+1;I (F3(76)-I)1.22
Ø1.18  S  SA=F3(I)-(<M*F1(I)>+F3(77))
Ø1.19  T  !,%4.Ø,F1(I),%8.Ø3,SA
Ø1.2Ø  I  (F1(SN)-I)1.4Ø,2.Ø1
Ø1.21  I  (I-F3(72))2.1Ø,2.2Ø,2.1Ø
Ø1.22  T  !,"AVG(",%2.Ø,F3(7Ø),")  ",%8.Ø3,QA/F3(7Ø)
Ø1.23  T  (QA/F3(7Ø))*(1ØØ/F3(74))
Ø1.24  A  !," RECALC, Y OR N?",Y
Ø1.25  I  (Y-ØY)1.26,1.41,1.24
Ø1.26  T  !,"DECL-",F3(74)
Ø1.27  T  !,"CONC S X DIL X UNITS FAC =",F3(73)
Ø1.28  T  !,"COMP/AVG TAB=",F3(75)
Ø1.29  T  !,"BL1=",F3(77),!,"BL2=",F3(79),!,"TIME=",F3(78)
Ø1.3Ø  A  !,"SPL NO",Q
Ø1.31  A  !,"DATE",Q
Ø1.32  A  !,"INITIALS",Q
Ø1.33  A  !,"PRODUCT",Q
Ø1.34  Q
Ø1.4Ø  T  !,"ERROR";Q
Ø1.41  L  G,%ONITA4,O

Ø2.Ø1  T  "   S";S SN=SN+1;GOTO 1.17
Ø2.1Ø  T  "   U"
Ø2.12  S  Q=<SA/ST>*F3(73)
Ø2.13  S  QA=QA+Q
Ø2.14  GOTO 2.23
Ø2.2Ø  T  "   C"
Ø2.21  DO 2.12
Ø2.22  S  Q=Q/F3(75)
Ø2.23  T  Q,<Q/F3(74)>*1ØØ
Ø2.24  GOTO 1.17
*
```

```
*L L,%ONITA4,Ø
*W A
C FOCAL-12

Ø1.Ø3  T   !,!,!;O T
Ø1.Ø4  L   O,F1,F,ITA,1
Ø1.Ø5  L   O,F3,F,ITA2,1
Ø1.Ø9  S   M=<F3(79)-F3(77)>/F3(78);
Ø1.14  T   !,"SECONDS   NET   PH   CODE   FOUND   % DECLARED",!,!
Ø1.16  S   I=Ø;S QA=Ø;S F3(81)=Ø
Ø1.17  S   I=I+1;I (F3(76)-I)1.22
Ø1.18  S   SA=F3(I)-(<M*F1(I)>+F3(77))
Ø1.19  T  !,%4.Ø,F1(I),%8.Ø3,SA
Ø1.2Ø  A   "      ",X;I (X-ØS)2.21,2.Ø1,1.21
Ø1.21  I   (X-ØX)2.11,1.17
Ø1.22  T   !,"AVG(",%2.Ø,F3(81),")          ",%8.Ø3,QA/F3(81)
Ø1.23  T   (QA/F3(81))*(1ØØ/F3(74))
Ø1.24  A   !,!,!,"SPL. NO.",Q
Ø1.25  A   !,"DATE",Q
Ø1.26  A   !,"INITIALS",Q
Ø1.27  A   !,"PRODUCT",Q
Ø1.29  T   !,!,!,!,!,!,!,!;Q

Ø2.Ø1  S   ST=SA;GOTO 1.17
Ø2.11  S   F3(81)=F3(81)+1
Ø2.12  S   Q=<SA/ST>*F3(73)
Ø2.13  S   QA=QA+Q
Ø2.14  GOTO 2.23
Ø2.21  DO 2.12
Ø2.22  S   Q=Q/F3(75)
Ø2.23  T   Q,<Q/F3(74)>*1ØØ
Ø2.24  GOTO 1.17
*
```



ABSORBANCE*

NOTE: Absorbance is converted to volts DC by the retransmitting slidewire connected to recorder pen slidewire pulley, and DC volts are in fact used for calculations.

DATA STORAGE SCHEMATIC

FILE F1

LOCATION	CONTENT
Ø	a/d ch. no.
1	x-coord. peak1
2	x-coord. peak2
3	x-coord. peak3
4	x-coord. peak4
5	x-coord. peak5
.	etc.
.	etc.
.	
.	
7Ø	# of stds.
71	first std cup #
72	2nd std cup #
73	3rd std cup #
74	etc.
75	etc.
76	etc.
77	etc.
78	"
79	"
8Ø	"
81	"
82	"
83	"
84	"

FILE F2

LOCATION	CONTENT
Ø	(vacant)
1	y-coord. peak 1
2	y-coord. peak2
3	y-coord. peak3
4	y-coord. peak4
5	y-coord. peak5
.	etc.
.	etc.
.	
.	
7Ø	# ITAS (tablets)
71	# cups (total)
72	comp. cup #
73	(std conc) (dil fac) (1/units)
74	amt. declared
75	wt. of comp./avg. tab. wt.
76	# of peaks found by program
77	y-coord. BL1 (x=Ø by defn.)
78	x-coord. BL2.
79	y-coord. BL2.
8Ø	Std conc
81*	std conc and max. y value
82*	units and x-coord. of y max.
83*	comp. wt. and current y
84*	avg. tab. wt. and current x

*(See notes below.)

NOTES

1. Files are floating point format files. Each file location actually occupies three core locations.

2. File locations 81 - 84 are multi-purpose utility storage locations. The first usage indicates contents and purpose for %ONITA1 and the second usage indicates contents and purpose for %ONITA2.

SECONDS	NET PH	CODE	FOUND	% DECLARED
339	Ø.565	S		
459	Ø.581	S		
573	Ø.589	S		
693	Ø.562	U	47.966	95.931
819	Ø.513	U	43.8Ø6	87.612
936	Ø.567	U	48.369	96.738
1Ø53	Ø.524	U	44.7Ø9	89.418
1176	Ø.565	U	45.2Ø5	96.41Ø
129Ø	Ø.574	S		
1413	Ø.551	U	47.Ø4Ø	94.Ø8Ø
1539	Ø.571	U	48.751	97.5Ø3
1653	Ø.535	U	45.642	91.283
1773	Ø.549	U	46.886	93.771
1893	Ø.567	U	48.396	96.793
2Ø19	Ø.589	S		
2136	Ø.561	U	47.916	95.831
2253	Ø.554	U	47.3Ø8	94.616
2376	Ø.523	U	44.649	89.298
2493	Ø.56Ø	U	47.778	95.555
2613	Ø.548	U	46.753	93.5Ø6
2733	Ø.59Ø	S		
2853	Ø.554	U	47.273	94.546
2976	Ø.556	U	47.466	94.933
3Ø9Ø	Ø.568	U	48.476	96.953
3216	Ø.591	U	5Ø.455	1ØØ.9Ø9
333Ø	Ø.536	U	45.744	91.487
3453	Ø.578	S		
357Ø	Ø.564	U	48.132	96.263
369Ø	Ø.621	U	53.Ø45	1Ø6.Ø9Ø
3816	Ø.565	U	48.252	96.5Ø3
393Ø	Ø.518	U	44.258	88.516
4Ø53	Ø.55Ø	U	46.953	93.9Ø6
4176	Ø.585	S		
4287	Ø.481	U	41.Ø68	82.135
441Ø	Ø.556	U	47.432	94.864
4533	Ø.558	U	47.626	95.251
4653	Ø.585	U	49.987	99.974
477Ø	Ø.574	U	49.Ø12	98.Ø25
489Ø	Ø.6Ø3	C	51.457	1Ø2.914
5Ø13	Ø.594	S		
513Ø	Ø.597	S		
AVG(3Ø)			47.312	94.623

```
RECALC, Y OR N?:N
DECL=            5Ø.ØØØ
CONC S X DIL X UNITS FAC = 5Ø.ØØØ
COMP/AVG TAB=    1.ØØØ
BL1=-            Ø.382
BL2=-            Ø.387
TIME=         5985.ØØØ
SPL NO:EXHIBIT
DATE:9MAR72
INITIALS:AUTHORS
PRODUCT:EXHIBIT *
```

DATA ACQUISITION AND CALCULATION

1. Data is acquired in groups of ten a/d samplings, each taken at the interval specified in the user's response to T during initialization. These ten readings are summed and the average is then considered as the y coordinate of a data point for defining curve shape. The x coordinate is time elapsed since setting SNS 1.

2. A peak is defined as five consecutive positive first derivatives followed after an unspecified interval of time by five consecutive negative first derivatives. The first derivative is obtained by subtracting the y coordinate of the previous data point from the y coordinate of the current data point. It should be noted that a zero first derivative will break the chain of five consecutive first derivatives.

3. A peak maximum is defined as the largest value obtained between the five consecutive positive and the five consecutive negative first derivatives.

4. Only the x and y coordinates for the peaks, initial baseline and final baseline are stored. All other data are lost.

5. The initial and final baseline y coordinates are defined as the average of the y coordinates of the last two data points taken prior to changing the position of SNS 1. (Note: x coordinate for BL1 = Ø).

6. Results are calculated from the following equations:

(a) $$\text{Found} = \frac{\text{Abs}_U}{\text{Abs}_S} \times \text{Conc Std} \times \text{Dil} \times \frac{1}{\text{Units}}$$

Abs_U = Sample abs. - Baseline Correction

Abs_S = Standard abs. - Baseline Correction

Conc Std is expressed in mg/ml

Dil = Volume to which 1 tablet is diluted (mls.)

1/Units = 1/1 for declared in mg.
1/64.8 for declared in grains (grs)

(b) $$\%\ \text{Declared} = \frac{\text{Found}}{\text{Declared}} \times 100\%$$

7. Baseline correction is accomplished by calculating the slope for a line between BL 1 and BL2. (See Eq. (c)).

(c) $$\text{Slope (m)} = \frac{y_2 - y_1}{x_2 - x_1}$$

The slope is then applied in equation (d) to calculate the baseline correction for each peak.

(d) $y = mx + b$; Where y = Baseline correction
x = x coordinate of peak max.
b = BL1

8. The value used for standard absorbance (Abs.) is obtained differently in %ONITA3 and %ONITA4. %ONITA3 determines an average standard value composed of all the standards excluding the first two standards and the final standard. %ONITA4 calculates using an individual standard value which must precede the unknowns or composite. See also next section.

SEQUENCING OF SAMPLES, STANDARDS AND COMPOSITE

1. Storage is provided for a maximum of 70 peaks, however this does not mean 70 unknowns. Careful study of the program and enclosed storage schematic will reveal two important conventions which must be adhered to in sequencing samples, standards and the composite in the auto-analyzer.

a. %ONITA3 requires a minimum of 4 standards, since the first two and the last standard value are not used for calculating.

b. %ONITA4 assumes at least one standard precedes any samples or the composite.

In view of the above, it becomes apparent that for the programs to work properly the run must contain at least 4 standards, one of which must be located in the first cup.

2. Provision is made for only one composite.

3. The above are minimum requirements and must be met for both programs to operate properly.

4. Sequencing of samples, standards and the composite beyond these requirements is left to the user's discretion.

Restriction: DIAL-MS modified so that if only one DF32 is present, Set Sense Switch Ø. For all other configurations, operate DIAL-MS as usual.

OPERATING INSTRUCTIONS

1. Mount DIAL-MS tape containing program on tape unit Ø and formatted LINCtape on unit 1.

2. Start DIAL-MS in the usual manner:

A. LSW Ø7Ø1 RSW 731Ø
B. Depress IO PRESET
C. Depress DO
D. Depress START 2Ø when the tape movements stop

3. When cursor appears on scope, type "—>LO FOCALSNS,Ø ↵" to load modified version of FOCAL-12.

4. When TTY responds with an "*" type "L G,%ONITA1,Ø↵" to load the first segment of the program.

5. Once %ONITA1 is loaded the following display will begin. Answer all questions on TTY. Use decimal numbers. Avoid FOCAL-12 terminators during response. The program is formatted assuming the space bar will be used to terminate responses. Typing an "←" deletes everything to the left of the "←" within the response, up to the colon.

INITIAL DISPLAY

CH NO:	a/d channel
NO OF ITAS, EXCLUDE COMP:	# of tablets to be assayed
NO PEAKS, SPLS+STDS+COMP:	Total # of cups in run
COMP CUP NO:	Location of composite

NEW DISPLAY

NO OF STDS:		Total # of standards in run
STD NO	CUP NO	The number under std no
1	:	refers to standard no..
2	:	Reply should indicate in
etc.		which cup std. no. 1 is located and so forth
STD DATA O.K., Y OR N:		Review the above table for accuracy. If changes are to be made type "N". Type "Y" to continue.
STD CONC:		Express in mg/ml
DIL FAC:		Dilution of tabs & comp (ml)
DECLARED IN MG OR GR?:		Type MG or GR (grains) to specify reporting units
AMT DECLARED:		Theo. amount/tablet, same unit as former.
WT OF COMP:		Composite weight (grams)
AVG TAB WR:		Express in grams (Note: If no composite type "1.0" in response to Wt of Comp and Avg Tab Wt.)

TAPE MOVEMENT

T:	Specify time interval in seconds at which the a/d is to be sampled.*

* A time interval of 0.3 secs. provides 2 - 3 points to define peak y maximum when data is collected from an auto-analyzer running 20 - 30 samples per hour.

6. After tapes stop and sample and display of data begins, depress sense switch 1 (SNS 1) to store initial baseline (BL1) and to begin search for peaks. (Note: Prior to depressing sense switch 1 data is not examined for peaks.)

7. The computer will now collect data and analyze it for peaks until SNS 1 is returned to its original position. This stops data collection, stores an average y value for the last two points to be used as BL2, and outputs results.

8. When calculations are complete the user is then asked "RECALC,Y OR N?". The user should now verify codes and determine if he desires to recalculate the results. If a "Y" is typed, %ONITA4 is loaded, the first absorbance value is outputted, and the user must type the proper code "s" (standard), "U" (unknown), "C" (composite) or "X" (deletion). Calculations are based on the user's coding. Deletion causes an absorbance to be omitted from the calculations.

9. After calculations are complete, the computer types the initial user input and asks the following questions. None of the following information is used by the program, but FOCAL-12 rules still apply to user input.

SPL NO:
DATE:
INITIALS:
PRODUCT:

(Note: In order to comply with FOCAL-12 rules governing the use of terminators within Ask statements. Mar. 12, 1972 may be written 12Mar72.)

10. After product is identified and the space bar is struck, the program stops and control returns to FOCALSNS. To restart the program type "L G,%ONITA1,Ø" and proceed as before.